X-543-64-366

NASA TMX-55182

IBM 7094 PROGRAM FOR THE SIX-COIL PROBLEM

BY
HENRY MILLER

DECEMBER 1964

NASA

GODDARD SPACE FLIGHT CENTER
GREENBELT, MARYLAND

X-543-64-366

IBM 7094 PROGRAM FOR THE SIX-COIL PROBLEM

by

Henry Miller

CONTENTS

Page

Introduction 1

Description of the Problem and Program 1

Input 8

Operating and Modifying Instructions 8

Appendix A – Flow Chart of Program 9

Appendix B – Fortran Listing 13

Appendix C – Fortran Notation 23

Appendix D – Error Printout 28

Appendix E – Sample Problem 30

IBM 7094 PROGRAM FOR THE SIX-COIL PROBLEM

INTRODUCTION

One method of producing a uniform magnetic field with a very large volume of homogeneity is to use a six-circular-coil system. This report gives a brief description of the system, along with an analysis of its solution on the IBM 7094 computer.

SIX-COIL PROBLEM

A theoretical discussion of the problem is given in the 1963 final report of the Goddard Summer Workshop Program.[1]

Our problem is to solve the following five equations:

$$\left.\begin{aligned}
&f_{3,1} + y\,f_{3,2} + z\,f_{3,3} = 0\\
&f_{5,1} + ys\,f_{5,2} + zt\,f_{5,3} = 0\\
&f_{7,1} + ys^2\,f_{7,2} + zt^2\,f_{7,3} = 0\\
&f_{9,1} + ys^3\,f_{9,2} + zt^3\,f_{9,3} = 0\\
&f_{11,1} + ys^4\,f_{11,2} + zt^4\,f_{11,3} = 0
\end{aligned}\right\} \qquad (1)$$

where the f's are the derivatives of Legendre Polynomials and are defined as follows:

---

[1] M. Speiser and D. L. Waidelich, "The Six-Circular-Coil System" in "Final Report of the Goddard Summer Workshop Program in Measurement and Simulation of the Space Environment" Publication X-320-63-264, Goddard Space Flight Center, Greenbelt, Maryland, pages C113-C121.

$$\left.\begin{aligned}
f_{3,i} &= 5\,x_i^2 - 1 \\
f_{5,i} &= 21\,x_i^4 - 14\,x_i^2 + 1 \\
f_{7,i} &= 429\,x_i^6 - 495\,x_i^4 + 135\,x_i^2 - 5 \\
f_{9,i} &= 2431\,x_i^8 - 4004\,x_i^6 + 2002\,x_i^4 - 308\,x_i^2 + 7 \\
f_{11,i} &= 29393\,x_i^{10} - 62985\,x_i^8 + 46410\,x_i^6 - 13650\,x_i^4 + 1365\,x_i^2 - 21,
\end{aligned}\right\} \qquad (2)$$

where $i = 1, 3$

This gives us five equations with seven unknowns: x_1, x_2, x_3, y, z, s, and t. So that a finite number of solutions can be obtained, we shall assume values for x_2 and x_3. This gives us x_1, y, z, s, and t to find.

Initially we pick a trial value of x_1, and check to see if the first four equations are satisfied. When two satisfactory trial values have been found, an attempt is made to satisfy the fifth equation by an iterative process that will be explained later.

After a trial value of x_1 has been chosen (the method will be explained later) the first job is to find all the non-negative values of s that are no greater than some predetermined upper bound for s from the following formula.

$$F(s) = a_{66}\,s^6 + a_{65}\,s^5 + a_{64}\,s^4 + a_{63}\,s^3 + a_{62}\,s^2 + a_{61}\,s + a_{60} = 0, \qquad (3)$$

where

$$\left.\begin{aligned}
a_{66} &= -\,d_3\,e_3 \\
a_{65} &= c_2^2 - (d_2\,e_3 + d_3\,e_2) \\
a_{64} &= 2\,c_1\,c_2 - (d_1\,e_3 + d_2\,e_2 + d_3\,e_1) \\
a_{63} &= (c_1^2 + 2\,c_0\,c_2) - (d_0\,e_3 + d_1\,e_2 + d_2\,e_1 + d_3\,e_0) \\
a_{62} &= 2\,c_0\,c_1 - (d_0\,e_2 + d_1\,e_1 + d_2\,e_0) \\
a_{61} &= c_0^2 - (d_0\,e_1 + d_1\,e_0) \\
a_{60} &= -\,d_0\,e_0
\end{aligned}\right\}$$

and where

$$\left.\begin{aligned}
c_2 &= f_{3,1}\, f_{5,2}\, f_{5,3}\, f_{7,1}\, f_{7,3}\, f_{9,2} - f_{3,3}\, f_{5,1}^2\, f_{7,2}^2\, f_{9,3} \\
c_1 &= 2\, f_{3,3}\, f_{5,1}\, f_{5,2}\, f_{7,1}\, f_{7,2}\, f_{9,3} - f_{5,3}\, f_{7,3}\, (f_{3,1}\, f_{5,2}\, f_{7,2}\, f_{9,1} \\
&\quad + f_{3,2}\, f_{5,1}\, f_{7,1}\, f_{9,2}) \\
c_0 &= f_{3,2}\, f_{5,1}\, f_{5,3}\, f_{7,2}\, f_{7,3}\, f_{9,1} - f_{3,3}\, f_{5,2}^2\, f_{7,1}^2\, f_{9,3}
\end{aligned}\right\} \tag{5}$$

$$\left.\begin{aligned}
d_3 &= f_{7,2}\, f_{9,2}\, (f_{3,1}\, f_{5,3}^2\, f_{7,1} - f_{3,3}\, f_{5,1}^2\, f_{7,3}) \\
d_2 &= f_{3,3}\, f_{5,1}\, f_{5,2}\, f_{7,1}\, f_{7,3}\, f_{9,2} - f_{3,1}\, f_{5,3}^2\, f_{7,2}^2\, f_{9,1} \\
d_1 &= f_{3,3}\, f_{5,1}\, f_{5,2}\, f_{7,2}\, f_{7,3}\, f_{9,1} - f_{3,2}\, f_{5,3}^2\, f_{7,1}^2\, f_{9,2} \\
d_0 &= f_{7,1}\, f_{9,1}\, (f_{3,2}\, f_{5,3}^2\, f_{7,2} - f_{3,3}\, f_{5,2}^2\, f_{7,3})
\end{aligned}\right\} \tag{6}$$

$$\left.\begin{aligned}
e_3 &= f_{3,1}\, f_{5,1}\, (f_{5,2}\, f_{7,3}^2\, f_{9,2} - f_{5,3}\, f_{7,2}^2\, f_{9,3}) \\
e_2 &= f_{3,1}\, f_{5,2}\, f_{5,3}\, f_{7,1}\, f_{7,2}\, f_{9,3} - f_{3,2}\, f_{5,1}^2\, f_{7,3}^2\, f_{9,2} \\
e_1 &= f_{3,2}\, f_{5,1}\, f_{5,3}\, f_{7,1}\, f_{7,2}\, f_{9,3} - f_{3,1}\, f_{5,2}^2\, f_{7,3}^2\, f_{9,1} \\
e_0 &= f_{3,2}\, f_{5,2}\, (f_{5,1}\, f_{7,3}^2\, f_{9,1} - f_{5,3}\, f_{7,1}^2\, f_{7,3})
\end{aligned}\right\} \tag{7}$$

To solve our sixth-degree equation for s, we are using the Newton-Rapson Method. The basic formula for this method is

$$s_{n+1} = s_n - \frac{F(s_n)}{F'(s_n)} \tag{8}$$

where

n = iteration number

$$F(s) = a_{66}\, s^6 + a_{65}\, s^5 + a_{64}\, s^4 + a_{63}\, s^3 + a_{62}\, s^2 + a_{61}\, s + a_{60}$$

$$F'(s) = 6\, a_{66}\, s^5 + 5\, a_{65}\, s^4 + 4\, a_{64}\, s^3 + 3\, a_{63}\, s^2 + z\, a_{62}\, s + a_{61}$$

Before the iteration is started, we must find an interval that contains one root. This is done by substituting values of s into F(s) until a change of sign occurs. We start with s = 0 and increment s by .01 until s equals some intermediate upper bound (predetermined) and then increment s by .1 until s equals the predetermined upper bound. If no change of sign occurs by the time s equals the upper bound our trial value of x_1 is discarded, and another trial value is used. If F(s)= 0 in one of the substitutions, the value of s is considered to be a root of F(s). In this program the intermediate upper bound has been set at s = 5, and the final upper bound at s = 10. If different values are desired, the change can be made by changing the source program and reassembling.

Once two consecutive values of S that give opposite signs when substituted in F(s) are found, an initial estimate of the root must be made so that the iteration can be started. In making the estimate, we shall assume that F(s) is linear in the interval (a, b), so

$$s_0 = -\frac{F(a)\,(b-a)}{F(b) - F(a)} + a \tag{9}$$

Now, we go into the iterations where

$$s_{n+1} = s_n - \frac{F(s_n)}{F'(s_n)}, \qquad n = 0, 1, 2, 3, \cdots$$

This iterative process is continued until

$$\Delta s_n = \left| \frac{F(s_n)}{F'(s_n)} \right| \leq 10^{-8}$$

If this is not accomplished in ten iterations, it is assumed that there is no solution in this interval, and another interval is attempted.

After a solution has been found or after it has been ascertained that there is no solution in the interval, an attempt is made to find another interval that contains a root. This process of seeking intervals and roots is continued until the upper bound on s is reached.

When all of the s 's have been found, the next job is to find the values of t by the following formula

$$t_i = \frac{B_{1,i}}{B_{2,i}} \tag{10}$$

$$\left.\begin{aligned} B_{1,i} &= \begin{vmatrix} f_{5,3}\,(A_{1,i}\, f_{3,1} + f_{7,1}) & A_{1,i}\, f_{3,3}\, f_{5,1} \\ f_{7,3}\,(A_{2,i}\, f_{5,1} + f_{9,1}) & A_{2,i}\, f_{5,3}\, f_{7,1} \end{vmatrix} \\ B_{2,i} &= \begin{vmatrix} f_{5,1}\, f_{7,3} & A_{1,i}\, f_{3,3}\, f_{5,1} \\ f_{7,1}\, f_{9,3} & A_{2,i}\, f_{5,3}\, f_{7,1} \end{vmatrix} \end{aligned}\right\} \tag{11}$$

and where

$$\left.\begin{aligned} A_{1,i} &= \frac{s_i\,(f_{5,2}\, f_{7,1} - s_i\, f_{5,1}\, f_{7,2})}{f_{3,2}\, f_{5,1} - s_i\, f_{3,1}\, f_{5,2}} \\ A_{2,i} &= \frac{s_i\,(f_{7,2}\, f_{9,1} - s_i\, f_{7,1}\, f_{9,2})}{f_{5,2}\, f_{7,1} - s_i\, f_{5,1}\, f_{7,2}}, \qquad i = 1, 2, \cdots, \leq 6 \end{aligned}\right\} \tag{12}$$

Although all values of t are printed, only the positive values of t are used in the subsequent calculations. If there are no positive values of t , a note to this effect is printed, and the current trial value of x_1 is discarded as not being suitable (and another trail value of x_1 is substituted).

For each positive value of t that exists we shall find y and z from the following relations

$$\left(\frac{y}{z}\right)_i = -\frac{f_{3,3}\, f_{5,1} - t_i\, f_{3,1}\, f_{5,3}}{f_{3,2}\, f_{5,1} - s_i\, f_{3,1}\, f_{5,2}} \tag{13}$$

$$z_i = -\frac{f_{3,1}}{\left(\frac{y}{z}\right)_i f_{3,2} + f_{3,3}} \tag{14}$$

$$y_i = \left(\frac{y}{z}\right)_i z_i, \qquad i = 1, 2, \cdots, \leq 6 \tag{15}$$

We now have x_1, x_2, x_3, s , t , y, and z such that the first four equations are satisfied. Our next job is to satisfy the fifth equation—namely

$$f_{11,1} + y\, s^4\, f_{11,2} + z t^4\, f_{11,3} = 0$$

To do this we shall use an iterative process of linear interpolation (or extrapolation). We compute the fifth equation as some function $g(x_1)$ as shown below

$$g(x_1) = f_{11,1} + y\, s^4\, f_{11,2} + z t^4\, f_{11,3} \tag{16}$$

We then seek another suitable x_1, and compute another $g(x_1)$. The first estimate of the true x_1 is obtained by

$$x_1 = \frac{x_{1B}\, g(x_{1A}) - x_{1A}\, g(x_{1B})}{g(x_{1A}) - g(x_{1B})} \tag{17}$$

where

$$x_{1A} = \text{first acceptance } x_1$$

$$x_{1B} = \text{second acceptance } x_1$$

For the next approximation, x_{1B} becomes x_{1A}, $g(x_{1B})$ becomes $g(x_{1A})$, x_1 becomes x_{1B}, and another $g(x_1)$ is computed to become the new $g(x_{1B})$. This process is continued until

$$\left| 1 - \frac{x_{1,i}}{x_{1,i+1}} \right| \leq 10^{-7}, \qquad i = \text{iteration number}$$

or until a previously specified number of iterations has been made. When computing a new $g(x_1)$, new values of s, t, y, and z must be found because they are dependent upon x_1. Also, since each x_1 can produce more than one s, and hence, more than one t, y, and z. There can be more than one $g(x_1)$ for a given x_1; therefore, more than one computed x_1 can be obtained from a pair of trial x_1's. From this, we see that the process of keeping track of everything can become quite involved.

When all of the final x_1's have been found, the last thing to do is to compute the values of s, t, y, and z for each x_1. Also, in this last pass, four more sets of values are found for each x_1; these are

$$r_2 = \sqrt{s} \tag{18}$$

$$r_3 = \sqrt{t} \tag{19}$$

$$j_2 = \frac{y}{r_2^3}\left(\frac{1 - x_1^2}{1 - x_2^2}\right) \tag{20}$$

$$j_3 = \frac{z}{r_3^3}\left(\frac{1 - x_1^2}{1 - x_3^2}\right) \tag{21}$$

As a check these solutions are substituted in the left-hand side of the five equations, and the results are printed.

At the start of this report I said that a couple values of x_1 are chosen, but I didn't say how they were obtained. There are several methods that can be used. In this program, lower bound is chosen, and this is the first trial value of x_1. If this value doesn't work, a given increment is added to this value. This process is continued until either a satisfactory x_1 is found, or until a predetermined upper limit is reached. With this x_1 added to the given increment being the lower bound, this process is repeated for the second x_1. When the solutions have been obtained for the five equations, the entire process is repeated with the second acceptable trial x_1 being the lower bound. This goes on until the upper limit has been reached.

This program was written for the IBM 7094 using Fortran IV. The appendices show the actual program along with the Input/Output specifications. With slight modification, this program should also work on other machines that use FORTRAN IV.

INPUT

The two input data cards are punched as follows:

Card Number	Columns	Fortran symbol	Format
1	1-14	X2	F14.7
	15-20	blank	–
	21-34	X3	F14.7
	35-40	blank	–
	41-42	MAXRUN	I2
2	1-14	X1	F14.7
	15-20	blank	–
	21-34	XMAX	F14.7
	35-40	blank	–
	41-45	DELTAX	F5.3

For a description of the FORTRAN symbols, see Appendix C.

OPERATING AND MODIFYING INSTRUCTIONS

This program is designed to be run under version 8 of the IBM 7094 IBSYS system using all seven index registers.

As the program is written, the upper bound on the acceptable solutions of F(s) is s = 10. This value can be easily changed by changing the constant in the comparison in statement 320 (card number 11260) in the source deck and reassembling.

Appendix A

FLOW CHART OF PROGRAM

FLOW CHART OF PROGRAM



Note:
The numbers in parentheses joined with dotted lines are the statement numbers in the source program.
The numbers in brackets which are inside the statement boxes indicate the statement number the respective box is near.



IBC = IB + ICNTB
Calculate XC(NC)
ISL3:0
NC = NC + 1
ISL3 = 1
ISL4:0
XC(1) = XC(NC)
NC = 2
ISL4 = 1
NC:2000
IB:LTB(NB)
IB = IB + 1
IA:LTA(NA)
IA = IA + 1
NC = NC − 1
LINE:50
LINE = LINE + 2
IPAGE = IPAGE + 1
New Page
Print Heading
Line = 6
Print XA(NA), XB(NB), GA(ICNTA+1), GB(ICNTB+1), XC(ICNTC+1)
LTB(NB):1
IB = 2
IA = 1
ND = 2
LTA(NA):1
ND = 2
IA = 2
IB = 1
IAC = IA + ICNTA
LINE = LINE + 1
IBC = IB + ICNTB
NDC = ND + ICNTC
LINE : 56
Print GA(IAC)
GB(IBC), XC(NDC)
IPAGE = IPAGE + 1
New Page
Print Continuation
Heading
LINE = 6
Print XA(NA), XB(NB), GA(IAC), GB(IBC), XC(NDC)
IB:TB(NB)
ND = ND + 1
IB = IB + 1
IA:LTA(NA)
ND = ND + 1
IA = IA + 1
ICNTC = NC
NB:LMTB
NB = NB + 1
ICNTB = ICNTB + LTB(NB)
ISL5:0
LMTG = ICNTB + LTB(NB)
ISL5 = 1
NA:LMTA
NB = 1
NA = NA + 1
ICNTA = ICNTA + 1
LMTC = NC
I RUN:MAXRUN
Error Printout
IRUN = IRUN + 1
NB = 1
XA(NB) = XB(NB)
LTA(NB) = LTB(NB)
NB:LMTB
NB = NB + 1
LMTA = LMTB
IBC = 1
GA(IBC) = GB(IBC)
IBC:LMTG
IBC = IBC + 1
ISL5 = 0
NC = 1
XB(NC) = XC(NC)
NC:LMTC
NC = NC + 1
LMTB = LMTC
ISL3:0
ISL2 = 2

Appendix B

FORTRAN LISTING

```
$IBFTC CØIL6    LIST,REF
      DIMENSIØN XA(2000),XB(2000),XC(2000),GA(2000),GB(2000),XDUM(4),
     1F3(4),F5(4),F7(4),F9(4),F11(4),S(7),LT(2000),A1(7),A2(7),B1(7),
     2B2(7),T(7),YØZ(7),Z(7),Y(7),LTA(2000),R2(7),R3(7),J2(7),J3(7),
     3LTB(2000),CHECK(7,6)
      WRITE (3,10)
   10 FØRMAT(1H1/////////////////////////////1H ,40X,16HSIX-CØIL PRØBLEM/1
     1H1)
      IRUN=1
      LMTA=1
      LMTB=1
      ISL1=0
      ISL2=0
      ISL3=0
      ISL4=0
      ISL5=0
      READ (2,20) X2,X3,MAXRUN
   20 FØRMAT(2(F14.7,6X),I2)
C
C     FIND ACCEPTABLE X11 AND X12
C
   30 READ (2,40) X1,XMAX,DELTAX
   40 FØRMAT(2(F14.7,6X),F5.3)
      XB(1)=X1
      WRITE(3,110)X2,X3
  110 FØRMAT(1H1/////////////////1H0,30X,61HCØMPUTATIØN ØF SIX-CØIL PRØBL
     1EM WITH THE FØLLØWING INPUT DATA///1H ,40X,5HX2 = ,F14.7//1H ,40X,
     25HX3 = ,F14.7)
      IPAGE=0
  120 ICNTA=0
      ICNTB=0
      NA=1
      NB=1
      NC=1
  130 IPAGE=IPAGE+1
      WRITE (3,140) IPAGE
  140 FØRMAT(1H1,110X,5HPAGE ,I4)
      XINT=XB(NB)
      IF(ISL2-1)147,147,143
  143 WRITE (3,145) XINT
  145 FØRMAT(1H0,43X,26HFINAL CØMPUTATIØN FØR X1= ,F14.7///)
      GØ TØ 155
  147 WRITE (3,150) XINT
  150 FØRMAT(1H0,40X,33HINTERMEDIATE CØMPUTATIØN FØR X1= ,F14.7///)
C
C     LEGENDRE PØLYNØMIALS
```

```
C
  155 WRITE (3,160)
  160 FORMAT(1H ,20HLEGENDRE POLYNOMIALS)
      WRITE (3,170)
  170 FORMAT(1H0,1HI,14X,5HF3(I),20X,5HF5(I),20X,5HF7(I),20X,5HF9(I),20X
     1,6HF11(I))
      XDUM(1)=XINT**2
      XDUM(2)=X2**2
      XDUM(3)=X3**2
      DO 180 I=1,3
      F3(I)=5.*XDUM(I)-1.
      F5(I)=(21.*XDUM(I)-14.)*XDUM(I)+1.
      F7(I)=((429.*XDUM(I)-495.)*XDUM(I)+135.)*XDUM(I)-5.
      F9(I)=(((2431.*XDUM(I)-4004.)*XDUM(I)+2002.)*XDUM(I)-308.)*XDUM(I)
     1+7.
      F11(I)=((((29393.*XDUM(I)-62985.)*XDUM(I)+46410.)*XDUM(I)-13650.)
     1*XDUM(I)+1365.)*XDUM(I)-21.
  180 WRITE (3,190) I,F3(I),F5(I),F7(I),F9(I),F11(I)
  190 FORMAT(1H ,I1,9X,4(E15.8,10X),E15.8)
C
C     FIND C,D,E AND A FOR F(S) EQUATION
C
      C2=F3(1)*F5(2)*F5(3)*F7(1)*F7(3)*F9(2)-F3(3)*F5(1)**2*F7(2)**2
     1*F9(3)
      C1=2.*F3(3)*F5(1)*F5(2)*F7(1)*F7(2)*F9(3)-F5(3)*F7(3)*(F3(1)*F5(2)
     1*F7(2)*F9(1)+F3(2)*F5(1)*F7(1)*F9(2))
      C0=F3(2)*F5(1)*F5(3)*F7(2)*F7(3)*F9(1)-F3(3)*F5(2)**2*F7(1)**2
     1*F9(3)
      D3=F7(2)*F9(2)*(F3(1)*F5(3)**2*F7(1)-F3(3)*F5(1)**2*F7(3))
      D2=F3(3)*F5(1)*F5(2)*F7(1)*F7(3)*F9(2)-F3(1)*F5(3)**2*F7(2)**2
     1*F9(1)
      D1=F3(3)*F5(1)*F5(2)*F7(2)*F7(3)*F9(1)-F3(2)*F5(3)**2*F7(1)**2
     1*F9(2)
      D0=F7(1)*F9(1)*(F3(2)*F5(3)**2*F7(2)-F3(3)*F5(2)**2*F7(3))
      E3=F3(1)*F5(1)*(F5(2)*F7(3)**2*F9(2)-F5(3)*F7(2)**2*F9(3))
      E2=F3(1)*F5(2)*F5(3)*F7(1)*F7(2)*F9(3)-F3(2)*F5(1)**2*F7(3)**2
     1*F9(2)
      E1=F3(2)*F5(1)*F5(3)*F7(1)*F7(2)*F9(3)-F3(1)*F5(2)**2*F7(3)**2
     1*F9(1)
      E0=F3(2)*F5(2)*(F5(1)*F7(3)**2*F9(1)-F5(3)*F7(1)**2*F9(3))
      WRITE (3,200) D3,E3
  200 FORMAT(1H0,40X,5HD3 = ,E15.8,20X,5HE3 = ,E15.8)
      WRITE (3,210) C2,D2,E2
  210 FORMAT(1H ,5HC2 = ,E15.8,20X,5HD2 = ,E15.8,20X,5HE2 = ,E15.8)
      WRITE (3,220) C1,D1,E1
  220 FORMAT(1H ,5HC1 = ,E15.8,20X,5HD1 = ,E15.8,20X,5HE1 = ,E15.8)
```

```
      WRITE (3,230) C0,D0,E0
  230 FORMAT(1H ,5HC0 = ,E15.8,20X,5HD0 = ,E15.8,20X,5HE0 = ,E15.8)
      A66=-D3*E3
      A65=C2**2-(D2*E3+D3*E2)
      A64=2.*C1*C2-(D1*E3+D2*E2+D3*E1)
      A63=C1**2+2.*C0*C2-(D0*E3+D1*E2+D2*E1+D3*E0)
      A62=2.*C0*C1-(D0*E2+D1*E1+D2*E0)
      A61=C0**2-(D0*E1+D1*E0)
      A60=-D0*E0
      WRITE (3,240) A66,A65,A64,A63,A62,A61,A60
  240 FORMAT(1H0,6X,3HA66,16X,3HA65,16X,3HA64,16X,3HA63,16X,3HA62,16X,
     13HA61,16X,3HA60/1H ,6(E15.8,4X),E15.8)
C
C     SOLUTIONS OF F(S)
C
      A=0
      I=1
      FA=A60
      IF(FA) 270,260,270
  260 S(I)=FA
      GO TO 440
  270 FD=FA
      AD=A
      A=A+.01
  280 FA=((((((A66*A+A65)*A+A64)*A+A63)*A+A62)*A+A61)*A+A60
      IF(FA)300,260,290
  290 IF(FD)380,310,310
  300 IF(FD)310,310,380
  310 IF(A-5.)270,320,320
  320 IF(A-10.)330,340,340
  330 FD=FA
      AD=A
      A=A+.1
      GO TO 280
  340 IF(I-1)350,350,370
  350 IF(X1-XMAX)360,353,353
  353 WRITE (3,355)
  355 FORMAT(1H0,52HTHIS PROBLEM CAN NOT BE SOLVED WITH THE GIVEN VALUES
     1///)
      GO TO 1220
  360 WRITE (3,365) X1
  365 FORMAT(1H0,4HX1= ,F14.7,28H IS NOT A SATISFACTORY VALUE///)
      X1=X1+DELTAX
      XB(1)=X1
      GO TO 130
  370 LT(NB)=I-1
```

```
      GO TO 450
  380 S(I)=-FD*(A-AD)/(FA-FD)+AD
      IC=1
  390 H=((((((A66*S(I)+A65)*S(I)+A64)*S(I)+A63)*S(I)+A62)*S(I)+A61)*S(I)
     1+A60)/(((((6.*A66*S(I)+5.*A65)*S(I)+4.*A64)*S(I)+3.*A63)*S(I)+2.
     2*A62)*S(I)+A61)
      S(I)=S(I)-H
      IF(ABS(H)-.00000001)440,440,400
  400 IF(IC-10)410,420,420
  410 IC=IC+1
      GO TO 390
  420 WRITE (3,430) S(I),XA(NA),XB(NB)
  430 FORMAT(1H0,3HS= ,F14.7,32H CONVERGES TOO SLOWLY WHEN X11= ,F14.7,
     110H AND X12= ,F14.7//)
      GO TO 310
  440 I=I+1
      GO TO 310
  450 WRITE (3,460)
  460 FORMAT(1H0,1HI,11X,4HS(I),15X,5HA1(I),17X,5HA2(I),17X,5HB1(I),17X,
     15HB2(I),17X,4HT(I))
      LTD=LT(NB)
      DO 470 I=1,LTD
      A1(I)=S(I)*(F5(2)*F7(1)-S(I)*F5(1)*F7(2))/(F3(2)*F5(1)-S(I)*F3(1)
     1*F5(2))
      A2(I)=S(I)*(F7(2)*F9(1)-S(I)*F7(1)*F9(2))/(F5(2)*F7(1)-S(I)*F5(1)
     1*F7(2))
      B1(I)=F5(3)*(A1(I)*F3(1)+F7(1))*A2(I)*F5(3)*F7(1)-F7(3)*(A2(I)
     1*F5(1)+F9(1))*A1(I)*F3(3)*F5(1)
      B2(I)=F5(1)*F7(3)*A2(I)*F5(3)*F7(1)-F7(1)*F9(3)*A1(I)*F3(3)*F5(1)
      T(I)=B1(I)/B2(I)
  470 WRITE (3,480) I,S(I),A1(I),A2(I),B1(I),B2(I),T(I)
  480 FORMAT(1H ,I1,7X,F11.7,5(7X,E15.8))
      IF(ISL2-1)510,510,490
  490 WRITE (3,500)
  500 FORMAT(1H0,6X,4HT(J),15X,4HY(J),15X,4HZ(J),14X,5HR2(J),14X,5HR3(J)
     1,14X,5HJ2(J),14X,5HJ3(J))
      GO TO 530
  510 WRITE (3,520)
  520 FORMAT(1H0,1HJ,14X,4HT(J),20X,6HY/Z(J),20X,4HZ(J),21X,4HY(Z),21X,
     14HG(J))
  530 J=1
      I=1
  540 IF(T(I))550,550,610
  550 IF(I-LTD)560,570,570
  560 I=I+1
      GO TO 540
```

```
570 IF(J-1)590,590,580
580 J=J-1
    IF(ISL2-1)650,750,680
590 WRITE (3,600)
600 FØRMAT(1H0,23HNØ PØSITIVE T AVAILABLE)
    IF(ISL1-1)350,350,603
603 IF(LMTB-1)604,604,605
604 WRITE(3,365) XINT
    GØ TØ 713
605 IF(NB-LMTB)608,606,606
606 LMTB=LMTB-1
    GØ TØ 770
608 LMTB=LMTB-1
    DØ 609 I=NB,LMTB
609 XB(I)=XB(I+1)
    GØ TØ 130
610 YØZ(J)=-(F3(3)*F5(1)-T(I)*F3(1)*F5(3))/(F3(2)*F5(1)-S(I)*F3(1)
   1*F5(2))
    Z(J)=-F3(1)/(YØZ(J)*F3(2)+F3(3))
    Y(J)=YØZ(J)*Z(J)
    IF(ISL2-1)620,720,660
620 GA(J)=F11(1)+Y(J)*S(I)**4*F11(2)+Z(J)*T(I)**4*F11(3)
    WRITE (3,630) J,T(I),YØZ(J),Z(J),Y(J),GA(J)
630 FØRMAT(1H ,I1,9X,4(E15.8,10X),E15.8)
    IF(I-LTD)640,650,650
640 J=J+1
    I=I+1
    GØ TØ 540
650 LTA(NA)=J
    ISL2=1
    ISL1=1
653 IF(X1-XMAX)655,353,353
655 XA(1)=X1
    X1=X1+DELTAX
    XB(1)=X1
    GØ TØ 130
660 R2(J)=SQRT(S(I))
    R3(J)=SQRT(T(I))
    J2(J)=Y(J)*(1.-X1**2)/(R2(J)**3*(1.-X2**2))
    J3(J)=Z(J)*(1.-X1**2)/(R3(J)**3*(1.-X3**3))
    WRITE (3,670) T(I),Y(J),Z(J),R2(J),R3(J),J2(J),J3(J)
670 FØRMAT(1H ,6(E15.8,4X),E15.8)
    CHECK(J,1)=F3(1)+Y(J)*F3(2)+Z(J)*F3(3)
    SDUM=S(I)*Y(J)
    TDUM=T(I)*Z(J)
    CHECK(J,2)=F5(1)+SDUM*F5(2)+TDUM*F5(3)
```

```
      SDUM=SDUM*S(I)
      TDUM=TDUM*T(I)
      CHECK(J,3)=F7(1)+SDUM*F7(2)+TDUM*F7(3)
      SDUM=SDUM*S(I)
      TDUM=TDUM*T(I)
      CHECK(J,4)=F9(1)+SDUM*F9(2)+TDUM*F9(3)
      SDUM=SDUM*S(I)
      TDUM=TDUM*T(I)
      CHECK(J,5)=F11(1)+SDUM*F11(2)+TDUM*F11(3)
      IF(I-LTD)640,680,680
  680 IF(NB-LMTB)690,700,700
  690 NB=NB+1
      GØ TØ 130
  700 WRITE(3,703) (I,I=1,5)
  703 FØRMAT(1H0,21HVALIDITY ØF SØLUTIØNS//1H ,1HJ,5(8X,9HEQUATIØN ,I1,
     12X))
      DØ 705 I=1,J
  705 WRITE(3,707) I,(CHECK(I,IT),IT=1,5)
  707 FØRMAT(1H ,I1,5(5X,E15.8))
      WRITE (3,710) XINT,X2,X3
  710 FØRMAT(1H0,5HX1 = ,F14.7/1H ,5HX2 = ,F14.7/1H ,5HX3 = ,F14.7)
  713 IF(X1-XMAX)715,1240,1240
  715 IRUN=1
      LMTA=1
      LMTB=1
      ISL1=0
      ISL2=0
      ISL3=0
      ISL4=0
      ISL5=0
      XB(1)=X1
      GØ TØ 120
  720 JB=J+ICNTB
      GB(JB)=F11(1)+Y(J)*S(I)**4*F11(2)+Z(J)*T(I)**4*F11(3)
      WRITE (3,630) J,T(I),YØZ(J),Z(J),Y(J),GB(JB)
      IF(I-LTD)730,750,750
  730 IF(JB-2000)640,740,740
  740 WRITE (3,745)
  745 FØRMAT(1H0,30HHELP - I AM BEING SQUEEZED ØUT///)
      GØ TØ 1240
  750 IF(ISL1-1)753,753,755
  753 ISL1=2
      XB(1)=X1
  755 LTB(NB)=J
      IF(NB-LMTB)760,770,770
  760 NB=NB+1
```

```
      ICNTB=ICNTB+LTB(NB)
      GØ TØ 130
  770 IPAGE=IPAGE+1
      WRITE (3,140) IPAGE
      WRITE (3,780) IRUN
  780 FØRMAT(1H ,43X,32HCØMPUTED VALUES ØF X13 FØR PASS ,I2//1H ,5X,3HX1
     11,23X,3HX12,23X,6HG(X11),21X,6HG(X12),22X,3HX13)
C
C     FIND X13
C
      LINE=4
      ICNTA=0
      ICNTB=0
      ICNTC=0
      NC=1
  790 IA=1
  800 IAC=IA+ICNTA
      IB=1
  810 IBC=IB+ICNTB
      XC(NC)=(XB(NB)*GA(IA)-XA(NA)*GB(IB))/(GA(IA)-GB(IB))
      IF(ABS(1.-XB(NB)/XC(NC))-.0000001)840,820,820
  820 IF(ISL3)830,830,870
  830 NC=NC+1
      GØ TØ 870
  840 ISL3=1
      IF(ISL4)850,850,860
  850 ISL4=1
      XC(1)=XC(NC)
      NC=2
      GØ TØ 870
  860 NC=NC+1
  870 IF(NC-2000)880,740,740
  880 IF(IB-LTB(NB))890,900,900
  890 IB=IB+1
      GØ TØ 810
  900 IF(IA-LTA(NA))910,920,920
  910 IA=IA+1
      GØ TØ 800
  920 NC=NC-1
      IF(LINE-50)950,930,930
  930 IPAGE=IPAGE+1
      WRITE (3,140) IPAGE
      WRITE (3,940) IRUN
  940 FØRMAT(1H ,37X,32HCØMPUTED VALUES ØF X13 FØR PASS ,I2,12H - CØNTIN
     1UED//1H ,5X,3HX11,23X,3HX12,23X,6HG(X11),21X,6HG(X12),22X,3HX13)
      GØ TØ 960
```

```
 950 LINE=LINE+2
 960 WRITE (3,970) XA(NA),XB(NB),GA(ICNTA+1),GB(ICNTB+1),XC(ICNTC+1)
 970 FØRMAT(1H0,2(F14.7,12X),2(E15.8,12X),E15.8)
     IF(LTB(NB)-1)975,975,977
 973 ND=2
     IA=2
     IB=1
     GØ TØ 980
 975 IF(LTA(NA)-1)1070,1070,973
 977 ND=2
     IA=1
     IB=2
 980 IAC=IA+ICNTA
 990 LINE=LINE+1
     IBC=IB+ICNTB
     NDC=ND+ICNTC
     IF(LINE-56)1010,1010,1000
1000 IPAGE=IPAGE+1
     WRITE (3,140) IPAGE
     WRITE (3,940) IRUN
     LINE=6
     WRITE (3,970) XA(NA),XB(NB),GA(IAC),GB(IBC),XC(NDC)
     GØ TØ 1030
1010 WRITE (3,1020) GA(IAC),GB(IBC),XC(NDC)
1020 FØRMAT(1H ,52X,2(E15.8,12X),E15.8)
1030 IF(IB-LTB(NB))1040,1050,1050
1040 IB=IB+1
     ND=ND+1
     GØ TØ 990
1050 IF(IA-LTA(NA))1060,1070,1070
1060 IA=IA+1
     ND=ND+1
     GØ TØ 980
1070 ICNTC=NC
1080 IF(NB-LMTB)1090,1100,1100
1090 ICNTB=ICNTB+LTB(NB)
     NB=NB+1
     GØ TØ 790
1100 IF(ISL5)1110,1110,1120
1110 ISL5=1
     LMTG=ICNTB+LTB(NB)
1120 IF(NA-LMTA)1130,1140,1140
1130 ICNTA=ICNTA+1
     NA=NA+1
     NB=1
     GØ TØ 790
```

Appendix C

FORTRAN SYMBOLS

Fortran Symbol	Math Symbol	Description	Equation Number
A	b	Current trial solution for S(I)	3,9
A1 A2	$A_{1,i}$ $A_{2,i}$	Intermediate values in finding y and z	12
A60 A61 A62 A63 A64 A65 A66	A_{60} A_{61} A_{62} A_{63} A_{64} A_{65} A_{66}	Coefficients of F(s)	4
AD	a	Previous trail solution for S(I)	3,9
B1 B2	$B_{1,i}$ $B_{2,i}$	Intermediate values in finding y and z	11
C0 C1 C2	c_0 c_1 c_2	Intermediate values used in finding the coefficients of F(s)	5
CHECK		Value of left-hand side of equations in cluster 1	
D0 D1 D2 D3	d_0 d_1 d_2 d_3	Intermediate values used in finding the coefficients of F(s)	6
DELTAX.		Increment applied to x_1 when finding suitable trial values	

Fortran Symbol	Math Symbol	Description	Equation Number
E0 E1 E2 E3	e_0 e_1 e_2 e_3	Intermediate values used in finding the coefficients of F(s)	7
F3 F5 F7 F9 F11	$f_{3,i}$ $f_{5,i}$ $f_{7,i}$ $f_{9,i}$ $f_{11,i}$	Derivatives of the Legendre polynomials, $P_j(x)$, $j = 3, 5, 7, 9, 11$	2
FA	F(b)	The current value of F(s)	3,9
FD	F(a)	The previous value of F(s)	3,9
GA GB	$g(x_{1A})$ $g(x_{1B})$	Used in making the value of x satisfy the fifth equation in (1)	16,17
H	$F(s_n)/F'(s_n)$	Iterative value substracted from s in finding the solution of F(s)	8
I		Running index	
IA		Running index for GA in generating new x_1	
IAC		Actual location of GA in block	
IB		Running index for GB in generating new x_1	
IBC		Actual location in GB in block	
IC		Loop counter in finding solution for F(s) (up to 10)	
ICNTA		Lower bound for index of XA for given NA and NB	

Fortran Symbol	Math Symbol	Description	Equation Number
LINE		Line count on page (used in print control)	
LMTA		Total number of x_{1A} (or XA)	
LMTB		Total number of x_{1B} (or XB)	
LMTC		Total number of new x_1 (or XC)	
LMTG		Total number of values of GB generated	
LT		Number of values of S or GB for a given x_{11} or x_{12} (4 uses)	
LTA		Number of values of S or GB for a given x_{11} (2 uses)	
LTB		Number of values of S or GB for a given x_{12} (2 uses)	
LTD		Dummy limit for DØ Statement (to eliminate subscript)	
MAXRUN		The maximum number of passes allowed in finding new x_1 (read in)	
NA		Running index for XA	
NB		Running index for XB	
NC		Index to XC (new x_1)	
ND		Running index for XC (new x_1)	
NDC		Actual location of XC (new x_1)	
R2	r_2	Solved solutions with satisfactory x_1	18
R3	r_3		19

Fortran Symbol	Math Symbol	Description	Equation Number
ICNTB		Lower bound for index of XB for given NA and NB	
ICNTC		Lower bound for index of XC(NC) for a given NA and NB in printing	
IPAGE		Page number (used in printing)	
IRUN		The pass number (up to MAXRUN)	
ISL1		0: G_{11} branch 1: G_{12} branch 2: end G_{12} branch	sense light simulators
ISL2		0: finding G_{11} 1: finding G_{12} 2: making final computation	sense light simulators
ISL3		0: No satisfactory new x_1 has been found 1: A satisfactory new x_1 has been found	sense light simulators
ISL4		0: Normal 1: Counter reset when first satisfactory new x_1 has been found	sense light simulators
ISL5		0: First time around in finding new x_1 1: Additional times around in finding new x_1 – don't re-calculate LMTG	sense light simulators
IT		Index for CHECK subscript	
J		Index (general)	
J2	j_2	Solved solutions with satisfactory x_1	20
J3	j_3	Solved solutions with satisfactory x_1	21
JB		Running index for GB	

```
 1140 LMTC=NC
      IF(IRUN-MAXRUN)1150,1200,1200
 1150 IRUN=IRUN+1
C
C     SHIFT VALUES OVER FOR NEXT PASS
C
      DO 1160 NB=1,LMTB
      XA(NB)=XB(NB)
 1160 LTA(NB)=LTB(NB)
      DO 1170 IBC=1,LMTG
 1170 GA(IBC)=GB(IBC)
      ISL5=0
      DO 1180 NC=1,LMTC
 1180 XB(NC)=XC(NC)
      LMTB=LMTC
      IF(ISL3)120,120,1190
 1190 ISL2=2
      GO TO 120
C
C     ERROR PRINTOUTS AND ENDINGS
C
 1200 WRITE (3,1210) MAXRUN
 1210 FORMAT(1H0,34HTHIS PROBLEM WILL NOT CONVERGE IN ,I2,11H ITERATIONS
     1///)
      GO TO 713
 1220 WRITE (3,1230)
 1230 FORMAT(1H0,40HTHIS OCURRED AT THE START OF THE PROGRAM///)
 1240 RETURN
      END
```

Fortran Symbol	Math Symbol	Description	Equation Number
S	s_i	Solution of F(s)	
SDUM		Dummy value used in computing CHECK	
T	t_i	Solution of equation cluster 1	10
TDUM		Dummy value used in computing CHECK	
X1	x_1	Solution of equation cluster 1 (also, initial value of trial x_1 (read in))	
X2	x_2	Given solution of equation cluster 1 (read in)	
X3	x_3	Given solution of equation cluster 1 (read in)	
XA	x_{1A}	First acceptable trial x_1	
XB	x_{1B}	Second acceptable trial x_1	
XC	x_1	New value of x_1	
XDUM	x_i^2	Dummy value used in computing F3-F11	
XINT		Dummy x_1 used in computation	
XMAX		Maximum allowable for x_1 (read in)	
Y	y_i	Solution of equation cluster 1	15
YOZ	$(y/z)_i$	Intermediate value in finding Y and Z	13
Z	z_i	Solution of equation cluster 1	14

Appendix D

ERROR PRINTOUTS

As an aid to the user, a series of error printouts has been incorporated in the program to indicate where the problem has gone astray. A description of these printouts follow.

Statement Number	Test of Printout and its Meaning
353	THIS PROBLEM CANNOT BE SOLVED WITH THE GIVEN VALUES No satisfactory solution can be found with the range of solutions that are allowed when this occurs, the program terminates.
360 and 604	(x_1) IS NOT A SATISFACTORY VALUE The current trial value of X will not work because no solution of F(s) can be found. The program tries another X.
420	(s_i) CONVERGES TOO SLOWLY WHEN X11= (x_{11}) AND X12= (x_{12}) The value of s in question will not converge within ten iterations when using the Newton-Rapson method of finding the solutions of F(s). Another value of S is tried.
590	NO POSITIVE T AVAILABLE None of the solutions of F(s) will give a positive value of T. Therefore our x_1 is not a satisfactory value, and another x_1 is tried.

Statement Number	Test of Printout and its Meaning
740	HELP–I AM BEING SQUEEZED OUT
	An excess of 2000 values of x_{11}, x_{12}, x_{13}, $g(x_{11})$ or $g(x_{12})$ occurs. When this happens, the program terminates
1200	THIS PROBLEM WILL NOT CONVERGE IN (MAXRUN) ITERATIONS
	A satisfactory convergence of x_{13} will not occur for a given part of x_1's. Another x_1 is tried.
1220	THIS OCCURRED AT THE START OF THE PROGRAM
	The error encountered occurred before any satisfactory solution of x_{13} was found. The program terminates.

Appendix E

SAMPLE PROBLEM

For a test run, the program was run with the following input data:

X2	=	.5917002
X3	=	.8710000
MAXRUN	=	10
X1	=	.2000000
XMAX	=	.2150000
DELTAX	=	.001

The first two acceptable values of X1 were .2090000 and .2110000.

When the iterative interpolation (or extrapolation) was applied to these values of X1, we get

Iteration	New X1
1	.2095635
2	.2096869
3	.2098787
4	.2098228
5	.2098287
6	.2098290
7	.2098290

with

X1	=	.2098290
X2	=	.5917002
X3	=	.8710000,

we solved for R2, R3, J2, J3, and got

R2	=	1.003553
R2	=	.998336
J2	=	0.
J3	=	0.

Substituting these solutions into the five equations, we obtain

Equation 1:	$-.1117587 \times 10^{-7}$
Equation 2:	$.2607703 \times 10^{-7}$
Equation 3:	$.3725290 \times 10^{-8}$
Equation 4:	$.2980232 \times 10^{-6}$
Equation 5:	$.1546740 \times 10^{-3}$

Since equation 5 contains fourth-degree terms, any error will be exaggerated when checking.